



SAS Publishing



Using TrueType Fonts with Universal Printing in SAS® 9.1

The correct bibliographic citation for this manual is as follows: SAS Institute Inc. 2003. *Using TrueType Fonts with Universal Printing in SAS® 9.1*. Cary, NC: SAS Institute Inc.

Using TrueType Fonts with Universal Printing in SAS® 9.1

Copyright © 2003, SAS Institute Inc., Cary, NC, USA

All rights reserved. Produced in the United States of America. No part of this publication may be reproduced, stored in a retrieval system, or transmitted, in any form or by any means, electronic, mechanical, photocopying, or otherwise, without the prior written permission of the publisher, SAS Institute Inc.

U.S. Government Restricted Rights Notice. Use, duplication, or disclosure of this software and related documentation by the U.S. government is subject to the Agreement with SAS Institute and the restrictions set forth in FAR 52.227-19 Commercial Computer Software-Restricted Rights (June 1987).


SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, December 2003

SAS Publishing provides a complete selection of books and electronic products to help customers use SAS software to its fullest potential. For more information about our e-books, e-learning products, CDs, and hard-copy books, visit the SAS Publishing Web site at support.sas.com/pubs or call 1-800-727-3228.

SAS® and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.



CHAPTER

1

Using TrueType Fonts with Universal Printing in SAS 9.1

<i>Introduction to Using TrueType Fonts with Universal Printing in SAS 9.1</i>	1
<i>Locating the TrueType Font Files</i>	1
<i>Registering the TrueType Font Files with the SAS Registry</i>	2
<i>Registering TrueType Fonts for Microsoft Windows, UNIX, or OpenVMS Alpha</i>	2
<i>Registering TrueType Fonts for z/OS</i>	2
<i>Using the TrueType Fonts</i>	2
<i>Specifying the TrueType Fonts with the Print Dialog Box</i>	2
<i>Specifying TrueType Fonts with SAS Program Statements</i>	3
<i>Specifying a Font with the SYSPRINTFONT Option</i>	3
<i>Examples of Specifying Fonts and Printing International Characters</i>	4
<i>Specifying a Font with SAS/GRAPH</i>	4
<i>Specifying a Font with PROC PRINT</i>	4
<i>Specifying a Font with PROC PRINT and a User-Defined ODS Template</i>	5
<i>Printing International Characters</i>	7

Introduction to Using TrueType Fonts with Universal Printing in SAS 9.1

The SAS 9.1 GIF, PCL, and PDF Universal printers use the FreeType library to generate TrueType fonts for output in all of the operating environments that SAS software supports. The output can be created with the ODS PRINTER statement, the ODS PCL statement, the ODS PDF statement, the SAS Print dialogs, or with SAS/GRAPH procedures.

TrueType fonts are portable across operating environments and are always available in Microsoft Windows environments. A few TrueType fonts are included with some versions of UNIX X Windows.

TrueType fonts support a wide range of international characters. For samples of SAS code that use TrueType fonts, see .

Locating the TrueType Font Files

SAS supplies two TrueType fonts: SAS Monospace and SAS Monospace Bold, and you can use other TrueType fonts with SAS. Locate the directory that contains the TrueType font files, and then use the FONTREG procedure to register these font files in the SAS Registry.

In Microsoft Windows environments, TrueType fonts are usually located in one of the following directories `C:\WINNT\Fonts\` or `C:\WINDOWS\Fonts\`. For all other operating environments, contact your system administrator for the location of TrueType fonts.

When you add TrueType fonts to a z/OS system, the TrueType font file must be allocated as a sequential data set with a fixed block record format and a record length of 1.

When SAS is invoked, the FONTSLOC system option points to the location of the fonts that are supplied by SAS, which include the TrueType fonts SAS Monospace and SAS Monospace Bold. Use the following OPTIONS statement to display the location of the SAS fonts:

```
proc options option=fontsloc;
run;
```

Registering the TrueType Font Files with the SAS Registry

To use TrueType fonts with Universal Printing, use the FONTREG procedure to register the TrueType fonts in the SAS Registry. For information about the FONTREG procedure, see the *Base SAS Procedures Guide*.

Registering TrueType Fonts for Microsoft Windows, UNIX, or OpenVMS Alpha

Submit the following SAS program. The FONTPATH statement specifies the directory that contains the TrueType fonts, and `pathname` is the directory path of the TrueType fonts.

```
proc fontreg mode=all;
    fontpath 'pathname';
run;
```

Registering TrueType Fonts for z/OS

Use the FONTFILE statement instead of the FONTPATH statement to specify the fixed block sequential file that contains a TrueType font.

```
proc fontreg mode=all;
    fontfile 'filename';
run;
```

After you update the SAS Registry, you can use the SAS Registry Editor to view a list of the registered fonts by typing REGEDIT on the SAS command line or by submitting a PROC REGISTRY statement that specifies the location of the files. For example:

```
proc registry list startat='core\printing\freetype\fonts';
run;
```

Using the TrueType Fonts

Specifying the TrueType Fonts with the Print Dialog Box

After you have updated the SAS Registry, the TrueType fonts are available for use within SAS. The following process describes how to access the TrueType fonts interactively when Universal Printing is enabled:

- 1 Select



- 2 Select an appropriate printer, such as PDF or PCL5.
- 3 Click the **Properties** button.
- 4 Click the **Font** tab.

This window contains drop-down boxes for Font, Style, Weight, Size (in points), and Character Set.

- 5 Click the arrow on the right side of the Font box and scroll through the list of available fonts.

TrueType fonts are indicated by a three-character tag enclosed in angle brackets. For example, the TrueType font Arial Unicode MS will be listed as <ttf> Arial Unicode MS. The <ttf> tag makes the distinction between the different kinds of fonts with the same name, such as <ttf> Symbol and a Symbol font that resides on a physical printer. Fonts that do not have a <ttf> tag reside in the printer's memory. If you designate PDF as the printer, the font is in the Adobe Reader rather than on a physical printer.

- 6 Select the font that you want to use.
- 7 Click **OK** to return to the Print dialog box.
- 8 Click **OK** again to create your output.

Specifying TrueType Fonts with SAS Program Statements

To choose a printer that supports TrueType fonts when you're using SAS in a batch or line-mode environment, submit the following options statement:

```
options printerpath=printername;
```

printername can be a printer name such as PCL5, PCL5e, or GIF.

Then, submit your program code with the font that you want to use. For example, if you want to use the TrueType font Arial Unicode MS on a test TITLE statement, include the following line of code in your SAS program:

```
Title1 f="Arial Unicode MS" "text in Arial Unicode MS";
```

When you use the SAS/GRAPH graphics device drivers SASPRTC, SASPRTG, and SASPRTM, you can specify style and weight attributes by using the forward slash (/) as a delimiter.

```
Title1 f="Arial/Italic/Bold" "text in bold italic arial";
```

For ODS templates, the style and weight attributes are specified after the text size parameter. See for a complete example.

Note: You should use the <ttf> tag only when it is necessary, for example, to distinguish between a TrueType font and another type of font with the same name. Δ

Specifying a Font with the SYSPRINTFONT Option

The SYSPRINTFONT system option sets the font that you want to use for printing a SAS document. For example, you could use the SYSPRINTFONT system option to print your output in the Arial Unicode MS font to your printer by submitting the following OPTIONS statement:

```
options sysprintfont="Arial Unicode MS";
```

You can also use the SYSPRINTFONT system option to specify the weight and size of a font. For example, the following code specifies an Arial font that uses bold face, is italicized, and has a size of 14 points:

```
SYSPRINTFONT=("Arial" bold italic 14)
```

For more information, see “SYSPRINTFONT= System Option” in *SAS Language Reference: Dictionary*.

Examples of Specifying Fonts and Printing International Characters

Specifying a Font with SAS/GRAPH

The following example creates an output with a title in an Arial font that uses bold face and is italicized.

```
goptions reset=all device=sasprtc;
title1 color=black f="Arial/Italic/Bold" "Sample Title in Bold Italic Arial";
options printerpath=pdf;
proc gslide;
run;
quit;
```

Specifying a Font with PROC PRINT

The following example adds information to the SAS Registry for all of the fonts in the fonts directory of your operating environment.

If you are running SAS on Microsoft Windows, the fonts should be located in either **C:\WINDOWS\Fonts** or in **C:\WINNT\Fonts**. If you cannot find the fonts directory in your operating environment, you can submit either of the following FONTREG procedures to direct SAS to the correct directory:

```
proc fontreg;
fontpath "%sysget(systemroot)\fonts";
run;
```

or

```
proc fontreg;
fontpath "?CSIDL_FONTS";
run;
```

After the Registry is updated, a PDF file is created that uses four of these fonts. Before submitting the following example, change the font names to the fonts that are installed on your system and change the fontpath to the directory that contains your font files.

```
proc fontreg;
fontpath 'pathname';
run;

filename new 'print1.pdf';
options printerpath=(PDF new);
ods printer;
proc print data=sashelp.class;
title1 f='Arial Unicode MS' h=2 'TrueType Arial Unicode MS';
title2 f='Times New Roman' h=3 'Times New Roman';
title3 f='Comic Sans MS ' 'Comic Sans MS ';
footnote1 f='Arial' h=4 'Arial h=4';
run;
ods printer close;
```

Figure 1.1 PDF Output Using PROC PRINT

Obs	Name	Sex	Age	Height	Weight
1	Alfred	M	14	69.0	112.5
2	Alice	F	13	56.5	84.0
3	Barbara	F	13	65.3	98.0
4	Carol	F	14	62.8	102.5
5	Henry	M	14	63.5	102.5
6	James	M	12	57.3	83.0
7	Jane	F	12	59.8	84.5
8	Janet	F	15	62.5	112.5
9	Jeffrey	M	13	62.5	84.0
10	John	M	12	59.0	99.5
11	Joyce	F	11	51.3	50.5
12	Judy	F	14	64.3	90.0
13	Louise	F	12	56.3	77.0
14	Mary	F	15	66.5	112.0
15	Philip	M	16	72.0	150.0
16	Robert	M	12	64.8	128.0
17	Ronald	M	15	67.0	133.0
18	Thomas	M	11	57.5	85.0
19	William	M	15	66.5	112.0

Specifying a Font with PROC PRINT and a User-Defined ODS Template

The SAS PDF Universal Printer has three Microsoft Windows TrueType fonts that are mapped to the equivalent Adobe Type 1 font. This mapping produces a smaller file size, but it supports only the Latin 1 encoding. To use the TrueType fonts that support more encodings, you must change their aliases in the SAS registry. You must change the aliases only if your locale requires an encoding other than the Latin 1 encoding.

The following example creates a template of font styles, imports the template into the SAS Registry, and then produces a PDF file.

Before you submit the example, change the font names to the fonts that are installed on your system and change the fontpath to the directory that contains your font files.

First, create a text file named `alias.sasxreg` that contains the following code:

```
[CORE\PRINTING\ALIAS\FONTS\PDF]
"Courier New"      = "<ttf> Courier New"
"Times New Roman" = "<ttf> Times New Roman"
"Arial"           = "<ttf> Arial"
```

Then, submit the following code:

```
proc registry import='alias.sasxreg';
run;

filename out 'print2.pdf';

options printerpath=(pdf out);
```

```

proc template;
define style New_style / store = SASUSER.TEMPLAT;
parent = styles.printer;
replace fonts /
'docFont'           = ("Times New Roman", 12pt)
'headingFont'       = ("Arial, Helvetica", 10pt, bold)
'headingEmphasisFont' = ("Arial, Helvetica", 10pt, bold italic)
'TitleFont'         = ("Arial, Helvetica", 12pt, italic bold)
'TitleFont2'        = ("Arial, Helvetica", 11pt, italic bold)
'FixedFont'         = ("Arial, Helvetica", 11pt)
'BatchFixedFont'    = ("Arial, Helvetica", 6pt)
'FixedHeadingFont'  = ("Courier New, Courier", 9pt, bold)
'FixedStrongFont'   = ("Courier New, Courier", 9pt, bold)
'FixedEmphasisFont' = ("Courier New, Courier", 9pt, italic)
'EmphasisFont'      = ("Arial, Helvetica", 10pt, italic)
'StrongFont'         = ("Arial, Helvetica", 10pt, bold)    ;
end;
run;

ods printer style=New_style;
proc print data=sashelp.class;
title1 'Proc Print';
title2 'Templated ODS output';
run;
ods printer close;

```

Figure 1.2 PDF Output Using PROC PRINT and a User-Defined ODS Template

*Proc Print
Templated ODS output*

Obs	Name	Sex	Age	Height	Weight
1	Alfred	M	14	69.0	112.5
2	Alice	F	13	56.5	84.0
3	Barbara	F	13	65.3	98.0
4	Carol	F	14	62.8	102.5
5	Henry	M	14	63.5	102.5
6	James	M	12	57.3	83.0
7	Jane	F	12	59.8	84.5
8	Janet	F	15	62.5	112.5
9	Jeffrey	M	13	62.5	84.0
10	John	M	12	59.0	99.5
11	Joyce	F	11	51.3	50.5
12	Judy	F	14	64.3	90.0
13	Louise	F	12	56.3	77.0
14	Mary	F	15	66.5	112.0
15	Philip	M	16	72.0	150.0
16	Robert	M	12	64.8	128.0
17	Ronald	M	15	67.0	133.0
18	Thomas	M	11	57.5	85.0
19	William	M	15	66.5	112.0

Printing International Characters

The following example must be run with the UTF-8 server, and requires a TrueType font that contains the characters that are used. The table of character names and the associated codes can be found on the Unicode web site at <http://www.unicode.org/charts>.

Before you submit the example, change the font names to the fonts that are installed on your system and change the fontpath to the directory that contains your font files.

First, create a file that contains the following code, and use either REGEDIT or PROC REGISTRY to import it into the SAS registry.

```
[CORE\PRINTING\FREETYPE\FAMILY]
"Courier" = "<ttf> Courier New"
"Helvetica" = "<ttf> Arial Unicode MS"
"Symbol" = "<ttf> Symbol"
"Times" = "<ttf> Arial Unicode MS"
[CORE\PRINTING\PRINTERS\PDF\ADVANCED]
"FONT FAMILIES" =LINK:"\\CORE\PRINTING\FREETYPE\FAMILY"
```

Then, submit the following code.

```
%macro utf8chr(ucs2);
kcvf(&ucs2, 'ucs2b', 'utf8');
%mend utf8chr;
%macro namechar(name, char);
name="&name"; code=upcase("&char"); char=%utf8chr("&char"x); output;
%mend namechar;

data uft8char;
length name $40;
%namechar(Registered Sign, 00AE);
%namechar(Cent Sign, 00A2);
%namechar(Pound Sign, 00A3);
%namechar(Currency Sign, 00A4);
%namechar(Yen Sign, 00A5);
%namechar(Rupee Sign, 20A8);
%namechar(Euro Sign, 20AC);
%namechar(Dong Sign, 20AB);
%namechar(Euro-currency Sign, 20A0);
%namechar(Colon Sign, 20A1);
%namechar(Cruzeiro Sign, 20A2);
%namechar(French Franc Sign, 20A3);
%namechar(Lira Sign, 20A4);
%namechar(Mill Sign, 20A5);
%namechar(Naira Sign, 20A6);
%namechar(Peseta Sign, 20A7);
%namechar(Won Sign, 20A9);
%namechar(New Sheqel Sign, 20AA);
%namechar(Latin Capital Ligature Oe, 0152);
%namechar(Thai Character Mo Ma, 0E21);
%namechar(Hangul Choseong Pieup-sios-pieup, 1124);
%namechar(Hiragana Letter Ro, 308D);
%namechar(Katakana Letter Ho, 30DB);
%namechar(Arabic Letter Yeh With Hamza Above, 0626);
%namechar(Hebrew Letter Shin, 05E9);
%namechar(Cyrillic Capital Letter Zhe, 0416);
```

```
%namechar(Latin Capital Letter A With Ring Above, 00c5);
%namechar(Greek Capital Letter Theta, 0398);
%namechar(Dingbat Negative Circled Digit One, 2776);
%namechar(Subscript Two, 2082);
%namechar(Superscript Four, 2074);
%namechar(Trade Mark Sign, 2122);
%namechar(Ohm Sign, 2126);
%namechar(Roman Numeral Four, 2163);
%namechar(Vulgar Fraction Two Thirds, 2154);
%namechar(Volume Integral, 2230);
%namechar(Box Drawings Heavy Up And Right, 2517);
%namechar(Female Sign, 2640);
%namechar(Male Sign, 2642);
%namechar(Beamed Sixteenth Notes, 266C);
%namechar(Unassigned CJK Unified Ideograph, 5327);
%namechar(Cjk Compatibility Ideograph-f902, F902);
%namechar(Hebrew Letter Shin With Dagesh And Shin Dot, FB2C);
%namechar(Dagger, 2020);
%namechar(Double Dagger, 2021);
%namechar(Place Of Interest Sign, 2318);
%namechar(Yin Yang, 262F);
%namechar(Cyrillic Capital Letter Ze, 0417);
run;
ods pdf;
proc print; run;
ods pdf close;
```

Figure 1.3 PDF Output of International Characters

The SAS System			
Obs	name	code	char
1	Registered Sign	00AE	®
2	Cent Sign	00A2	¢
3	Pound Sign	00A3	£
4	Currency Sign	00A4	¤
5	Yen Sign	00A5	¥
6	Rupee Sign	20A8	₹
7	Euro Sign	20AC	€
8	Dong Sign	20AB	₫
9	Euro-currency Sign	20A0	₠
10	Colon Sign	20A1	₡
11	Cruzeiro Sign	20A2	₧
12	French Franc Sign	20A3	₣
13	Lira Sign	20A4	₺
14	Mill Sign	20A5	₥
15	Naira Sign	20A6	₦
16	Peseta Sign	20A7	₧
17	Won Sign	20A9	₩
18	New Sheqel Sign	20AA	₪
19	Latin Capital Ligature Oe	0152	Œ
20	Thai Character Mo Ma	0E21	๒
21	Hangul Choseong Pieup-sios-pieup	1124	ㅍ
22	Hiragana Letter Ro	308D	ろ
23	Katakana Letter Ho	30DB	ホ
24	Arabic Letter Yeh With Hamza Above	0626	ﻯ

The SAS System

1

Obs	name	code	char
25	Hebrew Letter Shin	05E9	שׁ
26	Cyrillic Capital Letter Zhe	0416	Ж
27	Latin Capital Letter A With Ring Above	00C5	Å
28	Greek Capital Letter Theta	0398	Θ
29	Dingbat Negative Circled Digit One	2776	⦿
30	Subscript Two	2082	₂
31	Superscript Four	2074	⁴
32	Trade Mark Sign	2122	™
33	Ohm Sign	2126	Ω
34	Roman Numeral Four	2163	IV
35	Vulgar Fraction Two Thirds	2154	⅔
36	Volume Integral	2230	∭
37	Box Drawings Heavy Up And Right	2517	└
38	Female Sign	2640	♀
39	Male Sign	2642	♂
40	Beamed Sixteenth Notes	266C	♯
41	Unassigned CJK Unified Ideograph	5327	𠄎
42	Cjk Compatibility Ideograph-f902	F902	𠄎
43	Hebrew Letter Shin With Dagesh And Shin	FB2C	שׂ
44	Dagger	2020	†
45	Double Dagger	2021	‡
46	Place Of Interest Sign	2318	Ⓜ
47	Yin Yang	262F	☯
48	Cyrillic Capital Letter Ze	0417	З